	Program 30:02 Cihazın okuma yazma hızını ayarlamak (Quick Basic)

	'Quick Basic QB/L QB.QLB komut satırı ile yüklenmelidir

'Ayni dizin içerisinde QB.BI dosyası bulunmalıdır

'$INCLUDE: 'qb.bi'

'$DYNAMIC

DIM oku(20), yaz(20)

cihaz = &HB0

'A0h=Master    B0h=Slave     <-DEĞİŞTİRİN
bar = &H1F0

'01F0h=Primer  0170h=Sekonder <-DEĞİŞTİRİN
GOSUB Yerles

'Okuyan ve yazan assembler rutinleri yerleştir.

DEF SEG = 0: sakla0 = PEEK(0): sakla1 = PEEK(1) : OUT bar + 6, cihaz

CLS : PRINT " *** C/DVD- Okuma/Yazma Hız Seçimi *** " : PRINT STRING$(35, 205)

INPUT "Okuma hızı (Kb/S)", oku

v = INT(oku / 256): oku$ = CHR$(v) + CHR$(oku - (v * 256))

INPUT "Yazma hızı (Kb/S)", yaz

v = INT(yaz / 256): yaz$ = CHR$(v) + CHR$(yaz - (v * 256))

INPUT "0-Kombine 1-CAV  =", cav: IF cav THEN cav = 1

paket$ = CHR$(&HBB) + CHR$(cav) + oku$ + yaz$ + STRING$(6, 0)

GOSUB paketgonder: GOSUB bekle1 : GOSUB hatabul:  PRINT "Hata kodu= "; hata$:

cikis: DEF SEG = 0: POKE 0, sakla0: POKE 1, sakla1: END ' Int 0 vektörünü restore et /çık

‘Bütün Quick Basic programlarda kullanılan sabit parçadır

‘Her satırının nasıl çalıştığı Ek-C de anlatılmıştır

‘-------------------Belleğe yerleş--------------------------------
Yerles:


‘Data porta/porttan okuyan/yazan 2 tane kısa assembler

'rutini belleğe koyar. Okuyan oku(0), yazan ise yaz(0) dır.

‘Programın başında bu diziler DİM oku(?), yaz(?) ile tanımlanmış olmalıdır

DATA &h60, &H1E, &hba,&hf7,1, &Hb8,0,0, &h8e,&hd8

DATA  &hed,&ha3,0,0, &H1F, &h61, &hcb,-1

DEF SEG = VARSEG(oku(0)): baz = VARPTR(oku(0)): x = 0

DO: READ a: POKE baz + x, a: x = x + 1: LOOP UNTIL a = -1

POKE baz + 3, (bar AND &HFF)    'Seçilen portu sonradan ilave et

'Şimdi yazan assembler rutini belleğe koyalım

DATA &h60, &H1E, &hba,&hf7,1, &Hb8,0,0, &h8e,&hd8

DATA  &h3e,&ha1,0,0,&hef, &H1F, &h61, &hcb,-1

DEF SEG = VARSEG(yaz(0)): baz = VARPTR(yaz(0)): x = 0

DO: READ a: POKE baz + x, a: x = x + 1: LOOP UNTIL a = -1

POKE baz + 3, (bar AND &HFF)    'Seçilen portu sonradan ilave et

RETURN

 ‘-------------------Data portu oku--------------------------------

oku:


'Bu rutin data portundan 2 bayt okur.

'2 baytı a ve b değişkenine verir

DEF SEG = VARSEG(oku(0)): CALL ABSOLUTE(VARPTR(oku(0)))

DEF SEG = 0: a = PEEK(0): b = PEEK(1): DEF SEG : RETURN

‘-------------------Data porta yaz--------------------------------

yaz:


'Bu rutin a ve b sayılarını (2 baytı)

'data portuna yazar

DEF SEG = 0: POKE 0, a: POKE 1, b

DEF SEG = VARSEG(yaz(0)): CALL ABSOLUTE(VARPTR(yaz(0)))

DEF SEG : RETURN

‘-------------------Komut gönder----------------------------------

komutgonder:

' komut değişkenindeki komutu cihaza gönderi

' BSY biti normale dönünceye kadar bekler

'Komutun tekrarı gerekirse 50 tekrar yapar

'Arıza bulursa okur (hata=hata)

sayac = 0: hata = 0

10 : OUT bar + 7, komut: GOSUB bekle: tekrar = 0

FOR x = 1 TO 50: a = INP(bar + 7)

IF (a AND &H20) AND (sayac < 100) THEN tekrar = 1: EXIT FOR

IF INP(bar + 7) AND 8 = 1 THEN EXIT FOR

NEXT: IF tekrar THEN sayac = sayac + 1: GOTO 10

IF INP(bar + 7) AND 1 THEN hata = INP(bar + 1)

RETURN

‘-------------------Sıfır Bekle-----------------------------------

bekle:


'BAR =primer/sekonder, Cihaz=master/slave cihazinin

'status portunu 100 defa yoklar mesgul olmadigi zaman

'geri döner. Meşgul ise arıza bildirir

hata = 0: FOR zz = 1 TO 100

IF (INP(bar + 7) AND 128) = 0 THEN EXIT FOR

NEXT: IF INP(bar + 7) AND 1 THEN hata = INP(bar + 1) : RETURN

‘-------------------Bir Bekle-------------------------------------

bekle1:


‘BAR7'nin 3üncü bitin 1 olmasını bekler

'olmuyorsa hata kodu ile döner

bir = 1: hata = 0: FOR x = 1 TO 100000

IF (INP(bar + 7) AND 8) THEN bir = 0: EXIT FOR

NEXT: IF bir OR (INP(bar + 7) AND 1) THEN hata = INP(bar + 1)

RETURN

‘-------------------Paket Gönder----------------------------------

paketgonder:

'Cihaza paket$ ile verilmiş olan paketi yollar

'Dönüşte hata değişkeni (varsa) hatayı verir.

komut = &HA0: GOSUB komutgonder: GOSUB bekle1: IF hata THEN RETURN

'Cihaz paketi bekliyor. 2 şerli gönderelim:

FOR x = 1 TO LEN(paket$) STEP 2

a = ASC(MID$(paket$, x, 1)): b = ASC(MID$(paket$, x + 1, 1))

GOSUB yaz: NEXT: RETURN

‘-------------------Hata Analiz----------------------------------

hatabul:

'Cihaza Op.Kodu=03h paketi gönderilir

'hata1, hata2, hata3 ve hata$ değişkeni döner

paket$ = CHR$(&H3) + STRING$(3, 0) + CHR$(&HFD) + STRING$(7, 0)

GOSUB paketgonder: GOSUB bekle1: GOSUB oku: GOSUB oku: hata1 = (a AND 15)

FOR x = 1 TO 5: GOSUB oku: NEXT: hata2 = a: hata3 = b

DO: GOSUB oku: LOOP UNTIL (INP(bar + 7) AND 8) = 0

hata$ = HEX$(hata1) + "-" + HEX$(hata2) + "-" + HEX$(hata3) : RETURN

Programın ekran çıktısı görülmektedir. Cihazın okuma ve yazma hızları CAV olarak tesadüfi sayılar girilmiştir. Daha sonra yokladığımızda cihazın bu değerleri kendisine en yakın hızlara yuvarladığı görülmüştür. Bu program derlendiğinde 36.5 Kb büyüklüğünde bir exe dosya oluşur


Programda DEĞİŞTİRİN yazan satırları sisteminizin gereği olarak değiştiriniz. Burada kullanıcının okuma yazma hızı istekleri hex sayılara dönüştürülüp, ters hex formatında olarak paketteki yerine yerleştirilmiştir. Op kodu BBh paketi yollanmış ve arıza kontrolu yapılmıştır. Disk cihazdan çıkarılıncaya kadar bu hız uygulaması sabit kalacaktır. Bu programı çalıştırınca ve cihazın hızını yeterince yavaşlatınca bir CD-ROM cihazının ne kadar yavaşlayabileceğini görüp hayret edeceksiniz. 


Bu pakete benzeyen ve daha detaylı ayar yapmayı sağlayan bir başka kardeş paket daha vardır: Op Kodu=B6h paketi
